
ODBC API Overview



The ODBC API is a programming interface to the Open Database Connectivity libraries and drivers. The ODBC API provides high performance data access and database independence. 

	The ODBC API functions completely bypass Jet when accessing data. The result is that you could easily expect a performance increase of two to three times when accessing a SQL Server or Oracle database with the ODBC API instead of Jet.

	When you write code to the ODBC API you are writing code that automatically adapts to a variety of database types. Because you are writing to the ODBC functions, and not to a particular database type, you can easily change your applications’ backend by selecting a new database driver.

	There are other high-speed data access options available to the Visual Basic programmer, such as VBSQL and RDO. The ODBC API provides performance comparable to VBSQL and is easier to use because it requires fewer API calls than required by VBSQL. RDO is easier to use than the ODBC API, and provides comparable performance when accessing a remote datasource. But, the ODBC API will almost always perform better than RDO when your datasource is a local Access database.



How to use the ODBC API

The ODBC API functions reside in a DLL called ODBC.DLL on 16-bit operating systems, and it is called ODBC32.DLL in 32-bit operating systems. These DLLs are the driver managers. You must declare each ODBC function you plan to use, and reference one of those two DLLs. The functions should be declared as such. Keep in mind, however, this is not an exhaustive list, there are more than twice as many ODBC API methods as are listed here: 



' The hand is a Long in Windows 95 & �' Windows NT

#If Win32 Then

	Declare Function SQLAllocEnv Lib "odbc32.dll" (env As Long) As Integer

   Declare Function SQLFreeEnv Lib "odbc32.dll" (ByVal env As Long) As Integer

   

	Declare Function SQLAllocConnect Lib "odbc32.dll" (ByVal env As Long, ldbc As Long) As Integer

Declare Function SQLConnect Lib "odbc32.dll" (ByVal ldbc As Long, ByVal Server As String, ByVal serverlen As Integer, ByVal uid As String, ByVal uidlen As Integer, ByVal pwd As String, ByVal pwdlen As Integer) As Integer

	Declare Function SQLDriverConnect Lib "odbc32.dll" (ByVal ldbc As Long, ByVal hWnd As Long, ByVal szCSIn As String, ByVal cbCSIn As Integer, ByVal szCSOut As String, ByVal cbCSMax As Integer, cbCSOut As Integer, ByVal f As Integer) As Integer

#Else

	Declare Function SQLAllocEnv Lib "odbc.dll" (env As Long) As Integer

	Declare Function SQLFreeEnv Lib "odbc.dll" (ByVal env As Long) As Integer

   

	Declare Function SQLAllocConnect Lib "odbc.dll" (ByVal env As Long, ldbc As Long) As Integer

	Declare Function SQLConnect Lib "odbc.dll" (ByVal ldbc As Long, ByVal Server As String, ByVal serverlen As Integer, ByVal uid As String, ByVal uidlen As Integer, ByVal pwd As String, ByVal pwdlen As Integer) As Integer

	Declare Function SQLDriverConnect Lib "odbc.dll" (ByVal ldbc As Long, ByVal hWnd As Integer, ByVal szCSIn As String, ByVal cbCSIn As Integer, ByVal szCSOut As String, ByVal cbCSMax As Integer, cbCSOut As Integer, ByVal f As Integer) As Integer

#End If



Note the use of the #If Win32 Then directive that allows me to declare both sets of ODBC API calls in my application, which will make it easier to port across 16- and 32-bit Windows. Also notice that the function SQLDriverConnect takes a parameter called hWnd, which is a reference or a handle to a window. Window handles in 32-bit Windows and 32-bit VB 4.0 are of the long data type rather than the integer data type as in 16-bit Windows, VB 3.0, and the 16-bit version of VB 4.0. 

	The most commonly used ODBC API functions are:



• SQLAllocEnv(): sets aside memory for an ODBC Environment handle.

• SQLAllocConnect (): sets aside memory for a connection handle.

• SQLDriverConnect(): connect to a data source such as SQL Server or Access.

• SQLAllocStmt(): sets aside memory for a SQL statement handle.

• SQLExecDirect(): sends the SQL statement to the driver.

• SQLFetch(): retrieves one row of data from the statement handle. Only retrieves one row at a time.

• SQLGetData(): retrieves one column (field) of data from the current row in the statement handle.

• SQLFreeStmt(): releases the memory used by the statement handle.

• SQLDisconnect(): logs out of the datasource.

• SQLFreeConnect(): releases the memory used by the connection.

• SQLFreeEnv(): releases memory used by the ODBC environment.



In addition to declaring ODBC API functions, you will also need to set up an ODBC datasource through Windows control panel. Your new datasource will reference a database specific driver that will tell the driver manager how to talk to the specific database. 

	The sample database application that we discuss in the column uses the ODBC API functions. The code excerpts are from that application.



ODBC Initialization

This ODBC call checks to see if the libraries are present and if we can load them into memory and establish a memory area for them to work in:



   

If SQLAllocEnv(lEnv) <> 0 Then

MsgBox "Unable to initialize ODBC API drivers!"

 	ODBCInit = False

End

End If

  



	This Call establishes a memory area where we can put login information that will be used to login  to our data source. The area for the login information is contained in the ldbc variable. Both lEnv and ldbc as long data types.



   

If SQLAllocConnect(lEnv, ldbc) <> 0 Then

MsgBox "Could not allocate memory for connection Handle!"

      	ODBCInit = False

      	' Free the Environment

      	iStatus = SQLFreeEnv(lEnv)

      	If iStatus = SQL_ERROR Then

         		MsgBox "Error Freeing Environment From ODBC Drivers"

       	End If

      	' Quit the Application

      	End

 End If



ODBC - Connect to Data Source

	The SQLDriverConnect function takes login information passed from the Login boxand held in the sConnect variable. The handle Screen.ActiveForm.hWnd is a long in 32-bit Windows.



SConnect = “DSN=PRODUCTS;UID=User;PWD=Password;APP=;DATABASE=PROD.MDB”



If SQLDriverConnect(ldbc, Screen.ActiveForm.hWnd, sConnect, Len(sConnect), sResult, Len(sResult),            iSize, 1) <= 0 Then

 	MsgBox "Could not establish connection to ODBC driver!"

 	Exit Function

End If



ODBC: Sending SQL Statement

	In order to populate the list box, these functions are called, which set aside memory for a SQL query and then pass the query to the ODBC driver:



If SQLAllocStmt(ldbc, lStmt) <> 0 Then

MsgBox "Could not allocate memory for a statement handle!"

 	ODBCQuerySubmit = False

Exit Function 

End If

   

If SQLExecDirect(lStmt, sSQL, Len(sSQL)) <> SQL_SUCCESS Then

     lRet = SQLError(glEnv, ldbc, lStmt, sSQLState, lErrNo, sErrorMsg, MAX_DATA_BUFFER, iLen)

     sMsg = "Error Executing SQL Statement”

     ODBCQuerySubmit = False

End If



ODBC - Retrieving Data



	Get the next row of data:



bPerform = SQLFetch(lStmt)



	Get a single field of data from the current row:



iStatus = SQLGetData(lStmt, iColumn, 1, sData, MAX_DATA_BUFFER, lOutLen)



	Functions like the ones listed here are used to populate this box.

ODBC - Closing the  SQL Statement

Clear the Statement handle:



bPerform = SQLFreeStmt(lStmt, SQL_DROP)



ODBC - Closing the Connection

Finally, these functions are called to release memory associated with the environment and the database. Log out of the datasource:



If ldbc <> 0 Then

      iStatus = SQLDisconnect(ldbc)

      If iStatus = SQL_ERROR Then

         MsgBox "Error logging out of data source!"

      End If

End If



	Release memory associated with the connections:



iStatus = SQLFreeConnect(ldbc)

If iStatus = SQL_ERROR Then

      MsgBox "Error unloading ODBC drivers!"

 End If



	Release memory for the ODBC environment



 If lEnv <> 0 Then

      iStatus = SQLFreeEnv(lEnv)

      If iStatus = SQL_ERROR Then

           MsgBox "Error Freeing Environment From ODBC Drivers"

      End If

 End If

